Math 1. Two-Hours Exam December 10, 2017.

JE/JKL 10.12.17

Problem 1

```
> restart;with(LinearAlgebra):
```

Given the inhomogeneous linear system of equations

```
> lign1:=x1-2*x2+3*x3=a^2+10*a-3;
```

$$lign1 := x1 - 2\,x2 + 3\,x3 = a^2 + 10\,a - 3$$

```
> lign2:=x1+2*x2-5*x3=a^2+3;
```

$$lign2 := x1 + 2\,x2 - 5\,x3 = a^2 + 3$$

where a is an arbitrary real number.

The coefficient matrix **A** and the right-hand side **b** is

```
> A,b:=GenerateMatrix([lign1,lign2],[x1,x2,x3]);
```

$$A, b := \begin{bmatrix} 1 & -2 & 3 \\ 1 & 2 & -5 \end{bmatrix}, \begin{bmatrix} a^2 + 10\,a - 3 \\ a^2 + 3 \end{bmatrix}$$

Question 1

```
> a:=1:
```

The linear system of equations is then

```
> lign1;
```

$$x1 - 2\,x2 + 3\,x3 = 8$$

```
> lign2;
```

$$x1 + 2\,x2 - 5\,x3 = 4$$

The augmented matrix **T** = [**A**| **b**] is

```
> T:=GenerateMatrix([lign1,lign2],[x1,x2,x3],augmented=true);
```

$$T := \begin{bmatrix} 1 & -2 & 3 & 8 \\ 1 & 2 & -5 & 4 \end{bmatrix}$$

that has the reduced row echelon form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & -1 & 6 \\ 0 & 1 & -2 & -1 \end{bmatrix}$$

The completely reduced linear system of equations is then

$x_1 - x_3 = 6$

$x_2 - 2\,x_3 = -1$

if we set $x_3 = t$ we get the complete solution in standard parametric form

$(x_1, x_2, x_3) = (6, -1, 0) + t(1, 2, 1)$, $t \in \mathbb{R}$.

With Maple we get

```
> x:=LinearSolve(T,free=t);
```

$$x := \begin{bmatrix} 6 + t_3 \\ -1 + 2\,t_3 \\ t_3 \end{bmatrix}$$

From this we read the complete solution in standard parametric form
$(x_1, x_2, x_3) = (6, -1, 0) + t(1, 2, 1)$, $t \in \mathbb{R}$.

Question 2

```
> a:='a':
> x0:=<-7,7,-1>;
```

$$x0 := \begin{bmatrix} -7 \\ 7 \\ -1 \end{bmatrix}$$

$\mathbf{x}_0 = (-7, 7, -1)$ is a solution to the system of equations $\Leftrightarrow \mathbf{A}\mathbf{x}_0 = \mathbf{b}$.

```
> A.x0=b;
```

$$\begin{bmatrix} -24 \\ 12 \end{bmatrix} = \begin{bmatrix} a^2 + 10\,a - 3 \\ a^2 + 3 \end{bmatrix}$$

From this we see that $\mathbf{A}\mathbf{x}_0 = \mathbf{b} \Leftrightarrow a^2 + 3 = 12$ and $a^2 + 10\,a - 3 = -24 \Leftrightarrow a^2 = 9$ and
$a^2 + 10\,a + 21 = 0 \Leftrightarrow$
$a = \pm\, 3$ and $a^2 + 10\,a + 21 = 0$.
By insertion we see that only $a = -3$ fulfills the quadratic equation $a^2 + 10\,a + 21 = 0$. Thus, $(x_1, x_2, x_3) = (-7, 7, -1)$ is a solution to the system of equations $\Leftrightarrow a = -3$.

Problem 2

```
> restart;with(LinearAlgebra):
```

Given the symmetric matrix

```
> A:=<<0,0,0,-1>|<0,0,-1,0>|<0,-1,0,0>|<-1,0,0,0>>;
```

$$A := \begin{bmatrix} 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \\ 0 & -1 & 0 & 0 \\ -1 & 0 & 0 & 0 \end{bmatrix}$$

Question 1

The characteristic polynomial for **A** in completely factorized form is

```
> P(lambda):=factor(Determinant(A-lambda*IdentityMatrix(4)));
```

$$P(\lambda) := (\lambda - 1)^2\,(\lambda + 1)^2$$

From this we read that the characteristic polynomial for **A** has the double roots 1 and -1 which means that **A** has the eigenvalues 1 and -1 both with the algebraic multiplicity 2.

Question 2

```
> g:=(x,y)->evalb(x[1]<y[1]):
```

```
> ev:=sort(Eigenvectors(A,output=list),g);
```

$$ev := \left[\left[-1, 2, \left\{\begin{bmatrix}1\\0\\0\\1\end{bmatrix}, \begin{bmatrix}0\\1\\1\\0\end{bmatrix}\right\}\right], \left[1, 2, \left\{\begin{bmatrix}-1\\0\\0\\1\end{bmatrix}, \begin{bmatrix}0\\-1\\1\\0\end{bmatrix}\right\}\right]\right]$$

Maple gives 2 orthogonal basis vectors for both E_{-1} and E_1. Since the two eigenvector spaces E_{-1} and E_1 are orthogonal, because **A** is symmetric, we get an orthonormal basis for $\mathbb{R}^4$ equipped with the ordinary scalar product by norming the four basis vectors shown.

```
> q1:=1/sqrt(2)*<1,0,0,1>:
> q2:=1/sqrt(2)*<0,1,1,0>:
> q3:=1/sqrt(2)*<-1,0,0,1>:
> q4:=1/sqrt(2)*<0,-1,1,0>:
```

If

```
> Q:=<q1|q2|q3|q4>;
```

$$Q := \begin{bmatrix} \frac{1}{2}\sqrt{2} & 0 & -\frac{1}{2}\sqrt{2} & 0 \\ 0 & \frac{1}{2}\sqrt{2} & 0 & -\frac{1}{2}\sqrt{2} \\ 0 & \frac{1}{2}\sqrt{2} & 0 & \frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} & 0 & \frac{1}{2}\sqrt{2} & 0 \end{bmatrix}$$

and

```
> Lambda:=DiagonalMatrix([-1,-1,1,1]);
```

$$\Lambda := \begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

then **Q** is orthogonal and

$$\mathbf{Q}^{-1}\mathbf{A}\,\mathbf{Q} = \mathbf{\Lambda} \Leftrightarrow \mathbf{A} = \mathbf{Q}\,\mathbf{\Lambda}\,\mathbf{Q}^{-1} = \mathbf{Q}\,\mathbf{\Lambda}\,\mathbf{Q}^{T}.$$

Check using Maple

```
> Transpose(Q).Q;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> A=Q.Lambda.Transpose(Q);
```

$$\begin{bmatrix} 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \\ 0 & -1 & 0 & 0 \\ -1 & 0 & 0 & 0 \end{bmatrix} = \begin{bmatrix} 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \\ 0 & -1 & 0 & 0 \\ -1 & 0 & 0 & 0 \end{bmatrix}$$

Question 3

We consider the linear map $f: \mathbb{R}^4 \rightarrow \mathbb{R}^4$, that has ${}_e\mathbf{F}_e = [\, {}_e f(\mathbf{e}_1)\ {}_e f(\mathbf{e}_2)\ {}_e f(\mathbf{e}_3)\ {}_e f(\mathbf{e}_4)\,] = \mathbf{A}$.

```
> eFe:=A;
```

$$eFe := \begin{bmatrix} 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \\ 0 & -1 & 0 & 0 \\ -1 & 0 & 0 & 0 \end{bmatrix}$$

Furthermore we consider the 2-dimensional subspace $U = \{\mathbf{u} \in \mathbb{R}^4 \mid f(\mathbf{u}) = -\mathbf{u}\}$ in $\mathbb{R}^4$.
It is seen that $U = E_{-1}$.

```
> v1:=<1/2,1/2,1/2,1/2>;
```

$$v1 := \begin{bmatrix} \frac{1}{2} \\ \frac{1}{2} \\ \frac{1}{2} \\ \frac{1}{2} \end{bmatrix}$$

Since

```
> ef('v1')=eFe.v1;
```

$$ef(v1) = \begin{bmatrix} -\frac{1}{2} \\ -\frac{1}{2} \\ -\frac{1}{2} \\ -\frac{1}{2} \end{bmatrix}$$

then $f(\mathbf{v}_1) = -\mathbf{v}_1$. From this it follows that $\mathbf{v}_1 \in U$.
Also, it is seen that
$\mathbf{v}_1 = \frac{1}{2}(1, 0, 0, 1) + \frac{1}{2}(0, 1, 1, 0)$, where (1, 0, 0, 1) and (0, 1, 1, 0) according to question 2 are two basis vectors for U.
From this it follows that $\mathbf{v}_1 \in U$.
$\mathbf{v}_2 = \frac{1}{2}(1, 0, 0, 1) - \frac{1}{2}(0, 1, 1, 0) = (\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, \frac{1}{2})$ is also a unit vector in U.
Since the two vectors $\mathbf{v}_1$ and $\mathbf{v}_2$ are orthogonal, then the set $(\mathbf{v}_1, \mathbf{v}_2)$ is an orthonormal basis for U.
Of course, we could also have used "Gram-Schmidt" with e.g. $\mathbf{u}_2 = (1, 0, 0, 1) \in U$:
$\mathbf{w}_2 = \mathbf{u}_2 - (\mathbf{u}_2 \cdot \mathbf{v}_1)\, \mathbf{v}_1$

$\mathbf{v}_2 =$

$$\frac{1}{|\mathbf{w}_2|}\,\mathbf{w}_2\,.$$

Problem 3

```
> restart;with(LinearAlgebra):
> assume(t,real):
> interface(showassumed=0):
```

In the vector space $\mathbb{C}^\infty(\mathbb{R},\mathbb{C})$ a 4-dimensional subspace U is given by its basis:
$a = (\cos(t), \sin(t), e^t\cos(t), e^{(1+i)t})$.

A linear map f: U→U is given by the expression

```
> f:=x->diff(x,t,t)-2*diff(x,t)+2*x;
```

$$f := x \to \frac{\partial^2}{\partial t^2}\, x - 2\left(\frac{\partial}{\partial t}\, x\right) + 2\,x$$

Question 1

The images of the two last basis vectors are

```
> f(exp(t)*cos(t));
```

$$0$$

```
> f(exp((1+I)*t));
```

$$0$$

Since $f(e^t\cos(t)) = 0$ for all $t \in \mathbb{R}$ and $f(e^{(1+i)t}) = 0$ for all $t \in \mathbb{R}$, then the two basis vectors $e^t\cos(t)$ and $e^{(1+i)t}$ belongs to the kernel for f.

Question 2

The images of the two first basis vectors

```
> f(cos(t));
```

$$\cos(t) + 2\sin(t)$$

```
> f(sin(t));
```

$$\sin(t) - 2\cos(t)$$

From this and question 1 we read that

```
> aFa:=<<1,2,0,0>|<-2,1,0,0>|<0,0,0,0>|<0,0,0,0>>;
```

$$aFa := \begin{bmatrix} 1 & -2 & 0 & 0 \\ 2 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

(Concerning e.g. the second column: $f(\sin(t)) = -2\cos(t) + \sin(t)$)

Question 3

The coordinate vector for $5\cos(t)$ with respect to the basis a is

```
> a_5cos:=<5,0,0,0>;
```

$$a\_5cos := \begin{bmatrix} 5 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

$f(x(t)) = 5\cos(t) \Leftrightarrow {}_a f(x(t)) = {}_a 5\cos(t) \Leftrightarrow {}_a\mathbf{F}_a\, {}_a x(t) = {}_a 5\cos(t)$.

```
> x:=LinearSolve(aFa,a_5cos,free=c);
```

$$x := \begin{bmatrix} 1 \\ -2 \\ c_3 \\ c_4 \end{bmatrix}$$

I.e. all solutions to the equation are

$x(t) = \cos(t) - 2\sin(t) + c_1 e^t \cos(t) + c_2 e^{(1+i)t}$, t$\in\mathbb{R}$, $c_1, c_2 \in \mathbb{C}$.

```
> f(cos(t)-2*sin(t)+c1*exp(t)*cos(t)+c2*exp((1+I)*t));
```

$$5\cos(t)$$

as expected.

Exercise 4

```
> restart;with(LinearAlgebra):with(plots):
```

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \begin{bmatrix} 4 & -10 \\ 2 & -5 \end{bmatrix} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, t \in \mathbb{R}.$$

Question 1

```
> A:=<<4,2>|<-10,-5>>;
```

$$A := \begin{bmatrix} 4 & -10 \\ 2 & -5 \end{bmatrix}$$

```
> g:=(x,y)->evalb(x[1]<y[1]):
> sort(Eigenvectors(A,output=list),g);
```

$$\left[\left[-1, 1, \left\{\begin{bmatrix} 2 \\ 1 \end{bmatrix}\right\}\right], \left[0, 1, \left\{\begin{bmatrix} \frac{5}{2} \\ 1 \end{bmatrix}\right\}\right]\right]$$

From this we read that all eigenvalues for the system matrix **A** are 0 and -1 with the corresponding eigenvector spaces $E_0 = \text{span}\left\{\begin{bmatrix} 5 \\ 2 \end{bmatrix}\right\}$ and $E_{-1} = \left\{\begin{bmatrix} 2 \\ 1 \end{bmatrix}\right\}$.

$\begin{bmatrix} 5 \\ 2 \end{bmatrix} \in E_0$ and $\begin{bmatrix} 2 \\ 1 \end{bmatrix} \in E_{-1}$ are linearly inddependent, since the corresponding eigenvectors are different.

The complete solution to the system of differential uquations is then

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = c_1 \begin{bmatrix} 5 \\ 2 \end{bmatrix} + c_2 e^{-t} \begin{bmatrix} 2 \\ 1 \end{bmatrix}, t \in \mathbb{R}, c_1, c_2 \in \mathbb{R}.$$

Or written out in Maple

```
> x1:=t->c1*5+c2*2*exp(-t):
> x2:=t->c1*2+c2*exp(-t):
> 'x[1](t)'=x1(t);
```

$$x_1(t) = 5\, c1 + 2\, c2\, e^{-t}$$

```
> 'x[2](t)'=x2(t);
```

$$x_2(t) = 2\, c1 + c2\, e^{-t}$$

Question 2

From the figure we read that $(x_1(0), x_2(0)) = (1,0)$.

Therefore, we have for the determination of the constants c_1 og c_2 the linear sytem of equations

$$\begin{bmatrix} x_1(0) \\ x_2(0) \end{bmatrix} = c_1 \begin{bmatrix} 5 \\ 2 \end{bmatrix} + c_2 \begin{bmatrix} 2 \\ 1 \end{bmatrix} = \begin{bmatrix} 5 & 2 \\ 2 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} 1 \\ 0 \end{bmatrix}.$$

with corresponding augmented matrix

```
> T:=<<5,2>|<2,1>|<1,0>>;
```

$$T := \begin{bmatrix} 5 & 2 & 1 \\ 2 & 1 & 0 \end{bmatrix}$$

that has the reduced row echelon form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 1 \\ 0 & 1 & -2 \end{bmatrix}$$

From this we read that

```
> c1:=1;
```

$$c1 := 1$$

and

```
> c2:=-2;
```

$$c2 := -2$$

The solution wished for is then

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = \begin{bmatrix} 5 \\ 2 \end{bmatrix} - 2e^{-t} \begin{bmatrix} 2 \\ 1 \end{bmatrix}, t \in \mathbb{R}.$$

Or written out in Maple

```
> 'x[1](t)'=x1(t);
```

$$x_1(t) = 5 - 4\, e^{-t}$$

```
> 'x[2](t)'=x2(t);
```

$$x_2(t) = 2 - 2\, e^{-t}$$

where $t \in \mathbb{R}$.

Since e^{-t} tends towards 0 for t tending towards infinity, then $x_1(t)$ tends towards 5 and $x_2(t)$ towards 2, as t tends towards infinity.

```
> P1:=plot(x1(t),t=0..10,thickness=3,color=blue):
> P2:=plot(x2(t),t=0..10,thickness=3,color=red):
> punkter:=pointplot([[0,1],[0,0]],symbol=solidcircle,
  symbolsize=15):
> display(P1,P2,punkter,scaling=constrained,view=0..10);
```



Question 3

Other non-constant solutions, that fulfills the limit values wished for, can be found by fixing $c_1 = 1$ and changing c_2 . E.g.

```
> c1:=1;c2:=1;
```

$$c1 := 1$$

$$c2 := 1$$

The solution wished for is then

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = \begin{bmatrix} 5 \\ 2 \end{bmatrix} + e^{-t} \begin{bmatrix} 2 \\ 1 \end{bmatrix}, t \in \mathbb{R}.$$

Or written out in Maple

```
> 'x[1](t)'=x1(t);
```

$$x_1(t) = 5 + 2\, e^{-t}$$

```
> 'x[2](t)'=x2(t);
```

$$x_2(t) = 2 + e^{-t}$$

where $t \in \mathbb{R}$.

```
> P1:=plot(x1(t),t=0..10,thickness=3,color=blue):
> P2:=plot(x2(t),t=0..10,thickness=3,color=red):
> display(P1,P2,scaling=constrained,view=0..10);
```